#-------------------------------------------------
#
# Project created by QtCreator 2013-10-04T20:18:59
#
#-------------------------------------------------

QT       += network gui

VERSION = 1.1.1.1

DEPLOYMENT.display_name = Keppe Messenger Framework

android: DEFINES += "SYS_ANDROID"
shared: win32: TARGET = libkmsgfw
static: win32: TARGET = kmsgfw
unix: TARGET = kmsgfw
TEMPLATE = lib

DEFINES += KMSGFW_LIBRARY

SOURCES += KMsgFw.cpp \
    KMsgFw10.cpp \
    KMsgFw10Q.cpp \
    KMsgFw11Q.cpp \
    KMsgFw111Q.cpp

HEADERS += KMsgFw.h \
    KMsgFw10.h \
    KMsgFw10Q.h \
    KMsgFw11Q.h \
    KMsgFw111Q.h \
    KMsgFw_shared.h

OTHER_FILES += app.rc \
    KMsgFw \
    KMsgFw10 \
    KMsgFw10Q \
    KMsgFw11Q \
    KMsgFw111Q

shared: macx: CONFIG += lib_bundle
shared: win32: RC_FILE = app.rc
shared: win32: RESOURCES += res.qrc
shared: win32: DEFINES += WIN32_SHARED

unix: !macx: includefiles.path = /opt/keppemessenger/include
unix: !macx: includefiles.files = $$PWD/../include/KMsgFw.h \
    $$PWD/../include/KMsgFw10.h \
    $$PWD/../include/KMsgFw10Q.h \
    $$PWD/../include/KMsgFw11Q.h \
    $$PWD/../include/KMsgFw111Q.h \
    $$PWD/../include/KMsgFw_shared.h \
    $$PWD/../include/KMsgFw \
    $$PWD/../include/KMsgFw10 \
    $$PWD/../include/KMsgFw10Q \
    $$PWD/../include/KMsgFw11Q \
    $$PWD/../include/KMsgFw111Q

unix: !macx: target.path = /opt/keppemessenger/lib
unix: !macx: INSTALLS += target includefiles
